WS 2021/2022

Functional Programming

Week 9 – Calendar Application, Scope, Modules

René Thiemann Philipp Anrain Marc Bußjäger Benedikt Dornauer Manuel Eberl
Christina Kohl Sandra Reitinger Christian Sternagel

Department of Computer Science

Last Lecture – Library Functions

```haskell
take, drop :: Int -> [a] -> [a]
splitAt :: Int -> [a] -> ([a], [a])

takeWhile, dropWhile :: (a -> Bool) -> [a] -> [a]
span :: (a -> Bool) -> [a] -> ([a], [a])

zipWith :: (a -> b -> c) -> [a] -> [b] -> [c]
zip :: [a] -> [b] -> [(a, b)]
unzip :: [(a, b)] -> ([a], [b])

words, lines :: String -> [String]
unwords, unlines :: [String] -> String

concatMap :: (a -> [b]) -> [a] -> [b]

($) :: (a -> b) -> a -> b
```

Last Lecture – List Comprehension

- list comprehension
  - shape: `[ (x,y,z) | x <- [1..n], let y = x ^ 2, y > 100, Just z <- f y]`
  - consists of guards, generators, local declarations
  - translated via `concatMap`
- examples

```
prime n = n >= 2 && null [ x | x <- [2 .. n - 1], n `mod` x == 0]

ptriples n = [ (x,y,z) |
  x <- [1..n], y <- [x..n], z <- [y..n], x^2 + y^2 == z^2]
```

Last Lecture – Printing a Calendar

- given a month and a year, print the corresponding calendar
- example: December 2021

```
 Mo Tu We Th Fr Sa Su
        1  2  3  4  5
  6  7  8  9 10 11 12
  ...
```

- we concentrate on printing, assuming machinery for construction

```
type Month   = Int
type Year    = Int
type Dayname = Int -- Mo = 0, Tu = 1, ..., So = 6
-- monthInfo returns name of 1st day in m. and number of days in m.
monthInfo :: Month -> Year -> (Dayname, Int)
```

Design and Functionality for Representing Character-Pictures

```haskell
type Height = Int
type Width = Int
type Picture = (Height, Width, [[Char]])

above :: Picture -> Picture -> Picture
stack :: [Picture] -> Picture

beside :: Picture -> Picture -> Picture
spread :: [Picture] -> Picture

tile :: [[Picture]] -> Picture
tile = stack . map spread
```

Finalizing the Calendar

Creating Pictures

- single ‘pixels’

```
pixel :: Char -> Picture
pixel c = (1, 1, [[c]])
```

- rows

```
row :: String -> Picture
row r = (1, length r, [r])
```

- blank

```
blank :: Height -> Width -> Picture
blank h w = (h, w, blanks)
  where
    blanks = replicate h (replicate w ' ')
```

Constructing a Month

- as indicated, assume function
  `monthInfo :: Month -> Year -> (Dayname, Int)`
  where daynames are 0 (Monday), 1 (Tuesday), . . .

```haskell
daysOfMonth :: Month -> Year -> [Picture]
daysOfMonth m y =
  map (row . rjustify 3 . pic) [1 - d .. numSlots - d]
  where
    (d, t) = monthInfo m y
    numSlots = 6 * 7  -- max 6 weeks * 7 days per week
    pic n = if 1 <= n && n <= t then show n else ""

rjustify :: Int -> String -> String
rjustify n xs
  | l <= n = replicate (n - l) ' ' ++ xs
  | otherwise = error ("text (" ++ xs ++ ") too long")
  where l = length xs
```

Tiling the Days

- `daysOfMonth` delivers list of 42 single pictures (of size 1×3)
- missing: layout + header for final picture (of size 7×21)

```
month :: Month -> Year -> Picture
month m y = above weekdays . tile . groupsOfSize 7 $ daysOfMonth m y
  where weekdays = row " Mo Tu We Th Fr Sa Su"

-- groupsOfSize splits list into sublists of given length
groupsOfSize :: Int -> [a] -> [[a]]
groupsOfSize n [] = []
groupsOfSize n xs = ys : groupsOfSize n zs
  where (ys, zs) = splitAt n xs
```

Printing a Month

- transform a `Picture` into a `String`
  ```
  showPic :: Picture -> String
  showPic (_, _, css) = unlines css
  ```
- show result of `month m y` as `String`
  ```
  showMonth :: Month -> Year -> String
  showMonth m y = showPic $ month m y
  ```
- display final string via `putStr :: String -> IO ()` to properly print newlines and drop double quotes
  ```
  > showMonth 12 2021
  " Mo Tu We Th Fr Sa Su\n          1  2  3  4  5\n  6 ..."
  > putStr $ showMonth 12 2021
   Mo Tu We Th Fr Sa Su
             1  2  3  4  5
    6  7  8  9 10 11 12
   13 14 15 16 17 18 19
   20 21 22 23 24 25 26
   27 28 29 30 31
  ```

Scope

Scope

- consider program (1 compile error)

```
radius = 15
area radius = pi^2 * radius

squares x = [ x^2 | x <- [0 .. x]]

length [] = 0
length (_:xs) = 1 + length xs

data Rat = Rat Integer Integer
createRat n d = normalize $ Rat n d where normalize ... = ...
```

- scope
  - resolve ambiguities
  - defines which names of variables, functions, types, ... are visible at a given program position
  - controlling scope to structure larger programs (imports / exports)

Scope of Names

```haskell
radius = 15
area radius = pi^2 * radius
```

- in the following we assume that `name_i` in the real code is always just `name` and the `_i` is used for addressing the different occurrences of `name`
- renamed Haskell program
  ```haskell
  radius_1 = 15
  area_1 radius_2 = pi_1^2 * radius_3
  ```
- scope of names in right-hand sides of equations
  - is `radius_3` referring to `radius_2` or `radius_1`?
  - what is `pi_1` referring to?
- rule of thumb for searching `name`: search inside-out
  - think of abstract syntax tree of expression
  - whenever you pass a **`let`**, **`where`**, **`case`**, or function definition where `name` is bound, then refer to that local name
  - if nothing is found, then search global function `name`, also in Prelude
- `radius_3` refers to `radius_2`, `pi_1` to `Prelude.pi`

Local Names in Case-Expressions

- general case: `case expr of { pat1 -> expr1; ...; patN -> exprN }`
  - each `patI` binds the variables that occur in `patI`
  - these variables can be used in `exprI`
  - the newly bound variables of `patI` bind stronger than any previously bound variables
- example Haskell expression

```
case xs_1 of                -- renamed Haskell expression
  [] -> xs_2
  (x_1 : xs_3) -> case xs_4 ++ ys_1 of
    [] -> ys_2
    (x_2 : xs_5) -> x_3 : xs_6 ++ ys_3
```

  - `x_3` refers to `x_2` (since `x_2` is further inside than `x_1`)
  - `xs_6` refers to `xs_5` (since `xs_5` is further inside than `xs_3`)
  - `xs_4` refers to `xs_3`
  - `xs_1`, `xs_2`, `ys_1`, `ys_2`, and `ys_3` are not bound in this expression (the proper references need to be determined further outside)

Local Names in Let-Expressions

```
let {
  pat1 = expr1; ...; patN = exprN;
  f1 pats1 = fexpr1; ...; fM patsM = fexprM
} in expr
```

- all variables in `pat1` ... `patN` and all names `f1` ... `fM` are bound
- these can be used in `expr`, in each `exprI` and in each `fexprJ`
- variables of `patsJ` bind strongest, but only in `fexprJ`

- ```
  let (x_1, y_1) = (y_2 + 1, 5)     -- renamed Haskell expression
      f_1 x_2 = x_3 + g_1 y_3 id_1
      g_2 y_4 f_2 = f_3 $ g_3 x_4 f_4
  in (f_5, g_4, x_5, y_5)
  ```
  - `y_2`, `y_3` and `y_5` refer to `y_1`
  - `x_3` refers to `x_2` since `x_2` binds stronger than `x_1`
  - `x_4` and `x_5` refer to `x_1`
  - `f_3` and `f_4` refer to `f_2` since `f_2` binds stronger than `f_1`
  - `g_1`, `g_3` and `g_4` refer to `g_2`
  - `f_5` refers to `f_1`
  - `id_1` is not bound in this expression
  ```

Global Function Definitions

- general case:
  `fname pats = expr`
  - all variables in `pats` are bound locally and can be used in `expr`
  - `fname` is not locally bound, but added to global lookup table
  - all variables/names in `expr` without local reference will be looked up in global lookup table
  - lookup in global table does not permit ambiguities
-
  ```
  radius_1 = 15                           -- renamed Haskell program
  area_2 radius_2 = pi_1^2 * radius_3
  length_1 [] = 0
  length_2 (_:xs_1) = 1 + length_3 xs_2
  ```
  - `radius_1`, `area_2` and `length_1/2` are stored in global lookup table
  - global lookup table has ambiguity: `length_1/2` vs. `Prelude.length`
  - `pi_1` is not locally bound and therefore refers to `Prelude.pi`
  - `radius_3` refers to local `radius_2` and not to global `radius_1`
  - `xs_2` refers to `xs_1`
  - `length_3` is not locally bound and because of mentioned ambiguity, this leads to a compile error

Global vs. Local Definitions

```haskell
length :: [a] -> Int

-- choose definition 1,
length = foldr (\ _-> (1 +)) 0

-- definition 2,
length =
  let { length [] = 0; length (x : xs) = 1 + length xs }
  in length

-- or definition 3
length [] = 0
length (_ : xs) = 1 + length xs
```

- definitions 1 and 2 compile since there is no `length` in the rhs that needs a global lookup
- in contrast, definition 3 does not compile
- still definitions 1 and 2 result in ambiguities in global lookup table
  $\rightarrow$ study Haskell's module system

Modules

Modules

- so far
  - Haskell program is a single file, consisting of several definitions
  - all global definitions are visible to user

```
-- functions on rational numbers
data Rat = Rat Integer Integer          -- internal definition of datatype
normalize (Rat n d) = ...               -- internal function
createRat n d = normalize $ Rat n d     -- function for external usage
...
-- application: approximate pi to a certain precision
piApprox :: Integer -> Rat
piApprox p = ...
```

- motivation for modules
  - structure programs into smaller reusable parts without copying
  - distinguish between internal and external definitions
    - clear interface for users of modules
    - maintain invariants
    - improve maintainability

Modules in Haskell

```haskell
-- first line of file ModuleName.hs
module ModuleName(exportList) where
-- standard Haskell type and function definitions
```

- each `ModuleName` has to start with uppercase letter
- each module is usually stored in separate file `ModuleName.hs`
- if Haskell file contains no **`module`** declaration, ghci inserts module name `Main`
- `exportList` is comma-separated list of function-names and type-names, these functions and types will be accessible for users of the module
- if `(exportList)` is omitted, then everything is exported
- for types there are different export possibilities
  - `module Name(Type)` exports `Type`, but no constructors of `Type`
  - `module Name(Type(..))` exports `Type` and its constructors

Example: Rational Numbers

```
module Rat(Rat, createRat, numerator, denominator) where
data Rat = Rat Integer Integer
normalize = ...
createRat n d = normalize $ Rat n d
numerator (Rat n d) = n
...
instance Num Rat where ...
instance Show Rat where ...
```

- external users know that a type `Rat` exists
- they only see functions `createRat`, `numerator` and `denominator`
- they don't have access to constructor `Rat` and therefore cannot form expressions like `Rat 2 4` which break invariant of cancelled fractions
- they can perform calculations with rational numbers since they have access to `(+)` of class `Num`, etc., in particular for the instance `Rat`
- for the same reason, they can display rational numbers via `show`

Example: Application

```
module PiApprox(piApprox, Rat) where
-- Prelude is implicitly imported

-- import everything that is exported by module Rat
import Rat

-- or only import certain parts
import Rat(Rat, createRat)

-- import declarations must be before other definitions
piApprox :: Integer -> Rat
piApprox n = let initApprox = createRat 314 100 in ...
```

- there can be multiple `import` declarations
- what is imported is not automatically exported
  - when importing `PiApprox`, type `Rat` is visible, but `createRat` is not
  - if application requires both `Rat` and `PiApprox`, import both modules:

    ```
    import PiApprox
    import Rat
    ```

Resolving Ambiguities

```haskell
-- Foo.hs
module Foo where pi = 3.1415

-- Problem.hs
module Problem where

import Foo

pi = 3.1415
area r = pi * r^2
```

- problem: what is `pi` in definition of `area`? (global name)
- lookup map is ambiguous: `pi` defined in `Prelude`, `Foo`, and `Problem`
- ambiguity persists, even if definition is identical
- solution via qualifier: disambiguate by using `ModuleName.name` instead of `name`
  - write `area r = Problem.pi * r^2` in `Problem.hs` (or `area r = Prelude.pi * r^2`)

Qualified Imports

```haskell
module Foo where pi = 3.1415
module SomeLongModuleName where fun x = x + x

module ExampleQualifiedImports where

-- all imports of Foo have to use qualifier
import qualified Foo
-- result: no ambiguity on unqualified "pi"

import qualified SomeLongModuleName as S
-- "as"-syntax changes name of qualifier

area r = pi * r^2
myfun x = S.fun (x * x)
```

Summary

Summary

- calendar application
- scoping rules determine visibility of function names and variable names
- larger programs can be structured in modules
  - explicit export-lists to distinguish internal and external parts
  - advantage: changes of internal parts of module `M` are possible without having to change code that imports `M`, as long as exported functions of `M` have same names and types
  - if no module name is given: `Main` is used as module name
  - further information on modules
    `https://www.haskell.org/onlinereport/modules.html`